# Chapitre 1
# NOMBRES REELS ET APPROXIMATIONS

*Dans ce chapitre, nous abordons l'étude des suites et les différentes représentations des réels comme le développement en base b, ou en fractions continues.*
*La notion de vitesse de convergence ainsi que les techniques rapides de programmation se situent au cœur de notre réflexion, illustration avec diverses méthodes de calcul des décimales de $\pi$, y compris leur recherche isolée dans le système hexadécimal.*
*Nous expliquons en détails l'algorithme de calcul par Mathematica des constantes $e$ et $\pi$.*
*La constante d'Euler, présente dans tout système de calcul formel, se caractérise à partir de sa définition par une convergence fort lente. Nous exposons un algorithme peu connu, mais très performant d'accélération de convergence.*

# *I. SUITES DE NOMBRES REELS*

## 1. Généralités sur les suites

### 1.1 Définitions sous Mathematica

Il existe deux méthodes de programmation d'une suite récurrente $(u_n)_{n\in\mathbb{N}}$ définie par $u_0 = a$ et la relation $u_{n+1} = f(u_n)$ pour $n \in \mathbb{N}$ :
(1) *Définition récursive* en utilisant les fonctions à mémoire :

```
u[0]:=a
u[n_]:=u[n]=expr
```

(2) *Définition fonctionnelle* avec la primitive `Nest` :

```
u[n_]:=Nest[f,a,n]
```

Cette définition est préférable à la précédente, elle évite tous les inconvénients de la récursivité.
Dans la mesure du possible, introduire $f$ sous forme de fonction pure pour gagner en élégance, voire en efficacité.

*Définir sous Mathematica la suite $(u_n)_{n\in\mathbb{N}}$ définie par $u_0 = 1$ et $u_{n+1} = \sin(u_n)$ par les deux méthodes et comparer leur efficacité.*

```
u[0] := 1
u[n_] := u[n] = Sin[u[n - 1]]

Timing[N[u[200], 25]]

{0.01 Second, 0.1207876919408868316083668}

N[u[260]]

$RecursionLimit::reclim : Recursion depth of 256 exceeded.
```

Avec une définition récursive, l'appel de la fonction pour $n \geq 256$ provoque un avertissement. En effet, la variable système `$RecursionLimit` est par défaut égale à 256.
Pour modifier la valeur de cette variable, utilisons une structure de type `Block` à l'intérieur de laquelle on affecte *localement* une *valeur* supérieure, voire une valeur infinie.
A l'extérieur de la structure, la variable reprend sa valeur par défaut.

```
Clear[u]
u[0] := 1;
u[n_] := u[n] =
  Block[{$RecursionLimit = ∞}, Sin[u[n - 1]]]
```

Effectuons un test et vérifions que la variable `$RecursionLimit` conserve sa valeur globale 256. La structure de type `Block` est de ce point de vue fondamentalement différente de la structure définie par `Module` qui rend locales les variables par leurs noms.

```
{Timing[N[u[500], 25]], $RecursionLimit}

{{0.33 Second, 0.07698641344541407247897721}, 256}

Clear[v]
v[n_] := Nest[Sin, 1, n]

N[v[200], 25] // Timing

{0.09 Second, 0.1207876919408868316083668}
```

*On considère la suite* $(x_n)_{n\in\mathbb{N}}$ *définie par* $x_0 = 5$ *et* $x_{n+1} = x_n + \frac{1}{x_n}$.
*Déterminer un encadrement de* $x_{5000}$.

Dans cet exemple, nous approximons les termes de la suite par des nombres décimaux.
Entrons comme valeur initiale $x_0 = 5.0$ pour forcer Mathematica à effectuer des calculs avec des décimaux. En partant de la valeur intiale entière $x_0 = 5$, Mathematica n'effectuera pas d'approximations, le temps de calcul deviendra une fonction exponentielle par rapport à $n$.

```
Clear[u]
u[n_] := Nest[# + 1/# &, 5., n]

N[u[5000], 5] // Timing

{0.01 Second, 100.14}
```

En élevant au carré les deux membres de l'égalité $x_{k+1} = x_k + \frac{1}{x_k}$: $x_{k+1}^2 - x_k^2 = 2 + \frac{1}{x_k^2}$.

Sommons ces égalités pour $0 \le k \le n$, $\sum_{k=0}^{n} x_{k+1}^2 - \sum_{k=0}^{n} x_k^2 = 2\,(n+1) + \sum_{k=0}^{n} \frac{1}{x_k^2}$.

Donc $x_{n+1}^2 - x_0^2 = 2\,(n+1) + \sum_{k=0}^{n} \frac{1}{x_k^2}$. En conséquence,

$$x_{n+1}^2 - x_0^2 \ge 2\,(n+1).$$

Pour $n = 4999$ on obtient $x_{5000}^2 \ge 10025$, $x_{5000} \ge \sqrt{10025} = 100.125$.

Nous savons que ${x_{n+1}}^2 \geq {x_0}^2 + 2\,(n+1) > 2\,(n+1)$; donc pour $k \in \mathbb{N}^*$ $\frac{1}{{x_k}^2} \leq \frac{1}{2\,k}$.

Alors, ${x_{n+1}}^2 = {x_0}^2 + 2\,(n+1) + \sum_{k=0}^{n} \frac{1}{{x_k}^2} \leq {x_0}^2 + 2\,(n+1) + \frac{1}{{x_0}^2} + \frac{1}{2} \sum_{k=1}^{n} \frac{1}{k}$.

En particulier pour $n = 4999$, ${x_{5000}}^2 \leq 10025 + \frac{1}{25} + \frac{1}{2} \sum_{k=1}^{4999} \frac{1}{k}$.

Classiquement, par comparaison avec une intégrale, en utilisant la décroissance de la fonction $x \to \frac{1}{x}$ sur $\mathbb{R}^{+*}$ on démontre que $\sum_{k=1}^{n} \frac{1}{k} \leq 1 + \ln n$.

D'où l'inégalité : ${x_{5000}}^2 \leq 10025 + \frac{1}{25} + \frac{1}{2}\,(1 + \ln 5000) < 10030$.

Finalement : $100.125 < x_{5000} < 100.15$.

De cette étude nous déduisons que $x_n \underset{n \to +\infty}{\sim} \sqrt{2\,n}$.

Dans le cas d'une récurrence linéaire, nous utiliserons de préférence une troisième méthode souvent plus rapide à partir du produit matriciel (voir *Mathematica théorie et pratique, applications en algèbre*, algèbre linéaire, suites récurrentes).

## 1.2 Représentation graphique

Considérons la suite $(u_n)_{n \in \mathbb{N}}$ définie par $u_{n+1} = f(u_n)$ et $u_0 \in \mathbb{R}$.

Représentons graphiquement, en repère orthonormal, le graphe de $f$, la droite d'équation $y = x$ et les points de coordonnées $(u_k, f(u_k))_{0 \leq k \leq n-1}$ afin d'observer le comportement de la suite : monotonie, convergence.

Par exemple, $f(x) = \cos x$, $u_0 = 0.2$ et $n = 4$.

Représentons graphiquement les points dont les coordonnées sont définis par la liste

$\{\{u_0, 0\}, \{u_0, u_1\}, \{u_1, u_1\}, \{u_1, u_2\}, \{u_2, u_2\}, \{u_2, u_3\}, \{u_3, u_3\}, \{u_3, u_4\}, \{u_4, u_4\}, \{u_4, u_5\}\}$.

```
val = NestList[Cos, 0.2, 5]

{0.2, 0.980067, 0.556967, 0.848862, 0.660838, 0.789478}
```

Opérons sur les listes afin d'éviter toute programmation procédurale.

A partir de la liste *val* des itérées de $f$ : $(u_k)_{0 \leq k \leq 5}$.

- dupliquons les éléments de la liste *val* pour aboutir à la liste $l_1$
- supprimons de cette liste les deux premiers éléments et plaçons en tête le réel 0 : construction de la liste $l_2$
- supprimons de $l_1$ le dernier élément, et par transposition en adjoignant $l_2$, nous obtenons la liste finale.

```
l1 = Flatten[Map[{#, #} &, val]]
l2 = Prepend[Drop[l1, 2], 0]
```

```
{0.2, 0.2, 0.980067, 0.980067, 0.556967, 0.556967, 0.848862,
 0.848862, 0.660838, 0.660838, 0.789478, 0.789478}
```

```
{0, 0.980067, 0.980067, 0.556967, 0.556967, 0.848862,
 0.848862, 0.660838, 0.660838, 0.789478, 0.789478}
```

```
Transpose[{Drop[l1, -1], l2}]
```

```
{{0.2, 0}, {0.2, 0.980067}, {0.980067, 0.980067},
 {0.980067, 0.556967}, {0.556967, 0.556967},
 {0.556967, 0.848862}, {0.848862, 0.848862},
 {0.848862, 0.660838}, {0.660838, 0.660838},
 {0.660838, 0.789478}, {0.789478, 0.789478}}
```

Le graphique $g_1$ représente la fonction $f$ et la première bissectrice. Il est défini formellement et non visualisé grâce à l'option `DisplayFunction→Identity`.
Le graphique $g_2$ représente la liste de points.
$g_1$ et $g_2$ sont visualisés par la commande

`Show[g1,g2,DisplayFunction→$DisplayFunction]`.

```
g1 = Plot[{f[x], x}, {x, 0, 1.5},
   AspectRatio → Automatic, DisplayFunction → Identity];
g2 = Graphics[Line[Transpose[{Drop[l1, -1], l2}]]];
Show[g1, g2, DisplayFunction → $DisplayFunction];
```



Nous observons une suite non monotone qui semble converger vers la solution de l'équation $f(x) = x$ (à démontrer).

En généralisation, codons la fonction *plotSuites*, de paramètres :
la fonction $f$,
une liste du type $\{x_0, a, b\}$,
le nombre d'itérations.
$x_0$ est le premier terme de la suite, $[a, b]$ est l'intervalle sur lequel on représente $f$.

```
plotSuites[f_, {x0_, a_, b_}, n_] :=
 Module[{g1, g2, l1, l2, x},
  l1 := Flatten[Map[{#, #} &, NestList[f, x0, n]]];
  l2 = Prepend[Drop[l1, 2], 0];
  g1 = Plot[{f[x], x}, {x, a, b},
    AspectRatio → Automatic, DisplayFunction → Identity];
  g2 = Graphics[Line[Transpose[{Drop[l1, -1], l2}]]];
  Show[g1, g2, DisplayFunction → $DisplayFunction]]
```

*Application :* $f(x) = 3\sin(\frac{x}{4}) + \frac{4}{5}x$, $u_0 = \pi$. *Pour le tracé, fixons* $0 \le x \le 4\pi$ *et 12 itérations.*

```
plotSuites[3 Sin[#/4] + 4 #/5 &, {π, 0, 4 π}, 12];
```




Démontrer que la suite converge vers la solution de l'équation $\sin\frac{x}{4} = \frac{x}{15}$, $x \in [\pi, 4\pi]$.

```
N[FixedPoint[3 Sin[# / 4] + 4 # / 5 &, N[π, 50]], 50]

9.7396598857895357886897217696925649282737
```

## 2. Etude de la convergence des suites

Il convient de se montrer très critique dans la recherche de limites de suites en calcul formel. Les résultats diffèrent suivant les versions, peuvent être faux : la dernière version du logiciel n'étant pas nécessairement garante de fiabilité. Certaines limites de suites de ce chapitre étaient correctement explicitées dans la version 4, pas toujours dans la version 5, voire fausses. La version 6 assez lente, apporte des corrections, sans gommer toutes les erreurs.

### 2.1 Exemples avec Mathematica

*Soit $x \in \mathbb{R}$, étudier la convergence de la suite $(u_n)_{n\in\mathbb{N}}$ définie par* $u_n = \frac{1}{n^2} \sum_{k=1}^{n} E(k\,x)$.
*E désigne la fonction partie entière.*

Observons Mathematica (version 6) pour différentes valeurs de $x$ :

```
Clear[u]
u[n_, x_] := (1/n^2) Sum[Floor[k x], {k, 1, n}]
Limit[{u[n, 5], u[n, -7], u[n, 1/2], u[n, x]}, n → +∞]
```

$$\left\{\frac{5}{2}, -\frac{7}{2}, 0, \text{Limit}\left[\frac{\sum_{k=1}^{n} \text{Floor}[k\,x]}{n^2}, n \to \infty\right]\right\}$$

A partir de l'encadrement $k\,x - 1 \le E(k\,x) \le k\,x$, on déduit $\frac{1}{n^2} \sum_{k=1}^{n} (k\,x - 1) \le u_n \le \frac{1}{n^2} \sum_{k=1}^{n} k\,x$.
Laissons à Mathematica le mot de la fin :

```
Clear[v, w]
v[n_, x_] := (1/n^2) Sum[k x, {k, 1, n}]; w[n_, x_] := (1/n^2) Sum[(k x - 1), {k, 1, n}]
Limit[{v[n, x], w[n, x]}, n → +∞]
```

$$\left\{\frac{x}{2}, \frac{x}{2}\right\}$$

Par conséquent $\lim_{n\to+\infty} u_n = \frac{x}{2}$.
Donc le résultat affiché par Mathematica est faux pour $x = \frac{1}{2}$ (par exemple).

*Déterminer* $\lim\limits_{n\to+\infty} \sum\limits_{k=1}^{n} \sin\frac{k}{n}\,\sin\frac{k}{n^2}$ *(concours Mines-Ponts).*

Dans sa dernière version, Mathematica s'avère opérant (mais non dans la version 5.2, alors que l'on obtenait la limite avec Mathematica 4) :

```
Limit[Sum[Sin[k/n] Sin[k/n^2], {k, 1, n}], n → +∞]
```

```
-Cos[1] + Sin[1]
```

Expliquons rapidement ce résultat.

Définissons $v_n = \sum\limits_{k=1}^{n} \sin\frac{k}{n}\,\sin\frac{k}{n^2}$ et $u_n = \frac{1}{n}\sum\limits_{k=1}^{n} \frac{k}{n}\sin\frac{k}{n}$.

Le lecteur reconnaîtra dans $u_n$ une somme de Riemann associée à la fonction $x \to x\sin x$, la subdivision de l'intervalle [0, 1] étant $(\frac{k}{n})_{1\le k\le n}$.

Donc $\lim\limits_{n\to+\infty} u_n = \int_0^1 x\sin x\, dx = \sin 1 - \cos 1$.

Utilisons l'inégalité $0 \le x - \sin x \le \frac{x^3}{6}$, pour $x \ge 0$, que l'on applique à $x = \frac{k}{n^2}$.

$0 \le u_n - v_n = \sum\limits_{k=1}^{n} (\frac{k}{n^2} - \sin\frac{k}{n^2})\sin\frac{k}{n} \le \frac{1}{6n^6}\sum\limits_{k=1}^{n} k^3 \sin\frac{k}{n} \le \frac{1}{6n^6}\sum\limits_{k=1}^{n} k^3$.

La somme $\sum\limits_{k=1}^{n} k^3$ contient $n$ termes, le plus grand étant $n^3$. Donc $0 \le \sum\limits_{k=1}^{n} k^3 \le n\times n^3$, il en résulte que $\lim\limits_{n\to+\infty} u_n - v_n = 0$. Les deux suites $(u_n)_{n\in\mathbb{N}}$ et $(v_n)_{n\in\mathbb{N}}$ ont même limite.

$\lim\limits_{n\to+\infty} v_n = \sin 1 - \cos 1$.

```
u[n_] := Sum[k/n^2 Sin[k/n], {k, 1, n}]
{Limit[u[n], n → +∞], Integrate[x Sin[x], {x, 0, 1}]}
```

```
{-Cos[1] + Sin[1], -Cos[1] + Sin[1]}
```

*Etudier la suite* $(a_n)_{n\in\mathbb{N}}$ *définie par*
$a_0 = \frac{9}{2},\ a_1 = \frac{41}{9}$ *et* $a_{n+2} = 109 - \frac{920}{a_{n+1}} + \frac{2000}{a_n\, a_{n+1}}$.

```
Clear[a]
a[0] := 9/2; a[1] := 41/9
a[n_ /; n ≥ 2] := a[n] = 109 - 920/a[n - 1] + 2000/(a[n - 2] a[n - 1])
```